dfin_closure.mws

Math 262a, Fall 1999, Glenn Tesler
Sum/product/... of D-finite functions is D-finite
11/17/99

We have two functions defined by differential equations:

```
> 5*diff(f(x),x$2)+f(x)=0; f_sol := dsolve(",f(x));
  3*diff(g(x),x)+g(x)=0; g_sol := dsolve(",g(x));
```

$$5\left(\frac{\partial^2}{\partial x^2}\mathrm{f}(x)\right)+\mathrm{f}(x)=0$$

$$f\_sol := \mathrm{f}(x) = \_C1\cos\left(\frac{1}{5}\sqrt{5}\,x\right)+\_C2\sin\left(\frac{1}{5}\sqrt{5}\,x\right)$$

$$3\left(\frac{\partial}{\partial x}\mathrm{g}(x)\right)+\mathrm{g}(x)=0$$

$$g\_sol := \mathrm{g}(x) = \mathbf{e}^{(-1/3\,x)}\,\_C1$$

the constants for f, g need to be different, so rename it:

```
> g_sol := subs(_C1=_C3,g_sol);
```

$$g\_sol := \mathrm{g}(x) = \mathbf{e}^{(-1/3\,x)}\,\_C3$$

We want to find a differential equation satisfied by

```
> h(x) := -f(x) + g(x) + f(x)*g(x);
```

$$\mathrm{h}(x) := -\mathrm{f}(x)+\mathrm{g}(x)+\mathrm{f}(x)\,\mathrm{g}(x)$$

That is, for all settings of the constants _C1,_C2,_C3, the following will be a solution of whatever diffeq we find:

```
> h_sol := subs(f_sol,g_sol,h(x));
```

$$h\_sol := -\_C1\cos\left(\frac{1}{5}\sqrt{5}\,x\right)-\_C2\sin\left(\frac{1}{5}\sqrt{5}\,x\right)+\mathbf{e}^{(-1/3\,x)}\,\_C3 + \left(\_C1\cos\left(\frac{1}{5}\sqrt{5}\,x\right)+\_C2\sin\left(\frac{1}{5}\sqrt{5}\,x\right)\right)\mathbf{e}^{(-1/3\,x)}\,\_C3$$

We will find an equation solved by this without using this explicit solution. Just keep h(x) in terms of f(x), g(x), and derivatives, symollically, without ever plugging in the solutions obtained above.

All derivatives of h(x) involve derivatives of f(x), g(x) of various orders:

```
> diff(h(x),x);
```

$$-\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)+\left(\frac{\partial}{\partial x}\mathrm{g}(x)\right)+\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)+\mathrm{f}(x)\left(\frac{\partial}{\partial x}\mathrm{g}(x)\right)$$

```
> diff(h(x),x$2);
```

$$-\left(\frac{\partial^2}{\partial x^2}\mathrm{f}(x)\right)+\left(\frac{\partial^2}{\partial x^2}\mathrm{g}(x)\right)+\left(\frac{\partial^2}{\partial x^2}\mathrm{f}(x)\right)\mathrm{g}(x)+2\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\left(\frac{\partial}{\partial x}\mathrm{g}(x)\right)$$
$$+\mathrm{f}(x)\left(\frac{\partial^2}{\partial x^2}\mathrm{g}(x)\right)$$

and using the diffeqs satisfied by f and g, all 2nd order and higher derivatives of f can be replaced by some expression in f(x), diff(f(x),x), and all derivatives of g(x) can be replaced by multiples of g(x). Assuming we compute derivatives incrementally, this suffices to reduce the derivatives:

```
> myreduce := proc(fn)
     subs(diff(f(x),x$2)=-f(x)/5, diff(g(x),x)=-g(x)/3,
  fn)
  end:
```

Build a table of derivatives of h(x), reduced by the differential equations that f(x),g(x) satisfy.
The result will only involve f, df/dx, and g, and no other derivatives:

```
> dh[0] := h(x);
  for k from 1 to 6 do
        dh[k] := myreduce(diff(dh[k-1],x));
  od; k := 'k': # reset k to unevaluated symbol
```

$$dh_0 := -\mathrm{f}(x)+\mathrm{g}(x)+\mathrm{f}(x)\,\mathrm{g}(x)$$

$$dh_1 := -\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)-\frac{1}{3}\mathrm{g}(x)+\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)-\frac{1}{3}\mathrm{f}(x)\,\mathrm{g}(x)$$

$$dh_2 := \frac{1}{5}\mathrm{f}(x)+\frac{1}{9}\mathrm{g}(x)-\frac{4}{45}\mathrm{f}(x)\,\mathrm{g}(x)-\frac{2}{3}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)$$

$$dh_3 := \frac{1}{5}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)-\frac{1}{27}\mathrm{g}(x)+\frac{2}{15}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)+\frac{22}{135}\mathrm{f}(x)\,\mathrm{g}(x)$$

$$dh_4 := -\frac{1}{25}\mathrm{f}(x)+\frac{1}{81}\mathrm{g}(x)-\frac{164}{2025}\mathrm{f}(x)\,\mathrm{g}(x)+\frac{16}{135}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)$$

$$dh_5 := -\frac{1}{25}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)-\frac{1}{243}\mathrm{g}(x)-\frac{244}{2025}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)+\frac{4}{1215}\mathrm{f}(x)\,\mathrm{g}(x)$$

$$dh_6 := \frac{1}{125}\mathrm{f}(x)+\frac{1}{729}\mathrm{g}(x)+\frac{2096}{91125}\mathrm{f}(x)\,\mathrm{g}(x)+\frac{88}{2025}\left(\frac{\partial}{\partial x}\mathrm{f}(x)\right)\mathrm{g}(x)$$

Symbollically, all of the above are in the span of the following 5 elements:

```
> B_module := [ f(x), diff(f(x),x), g(x), f(x)*g(x),
  diff(f(x),x)*g(x) ];
```

$$B\_module := \left[f(x), \frac{\partial}{\partial x} f(x), g(x), f(x)\, g(x), \left(\frac{\partial}{\partial x} f(x)\right) g(x) \right]$$

(Note: it's possible that when we plug in the solutions for f(x), g(x), that the above would be dependent, but that's irrelevant. We treat these as formal, linearly independent symbols, and the true values of these give a space that's a "homomorphic image" of the symbollic space we work in.)

which implies that any 6 of these are linearly dependent (6>5), so we take dh[0]..dh[5].

Form a dependency equation. The symbollic form uses capital letters etc. to avoid having the computations actually be carried out, while in the other form they are carried out.

```
> dependency_symbollic := a[0]*‘ h‘(x) + sum(a[k]*Diff(‘
  h‘(x),x$k),k=1..5);
  dependency := sum(a[k]*dh[k],k=0..5);
```

$$dependency\_symbollic := a_0\ h(x) + a_1 \left(\frac{\partial}{\partial x}\ h(x)\right) + a_2 \left(\frac{\partial^2}{\partial x^2}\ h(x)\right) + a_3 \left(\frac{\partial^3}{\partial x^3}\ h(x)\right) + a_4 \left(\frac{\partial^4}{\partial x^4}\ h(x)\right) + a_5 \left(\frac{\partial^5}{\partial x^5}\ h(x)\right)$$

$$\begin{aligned} dependency := {} & a_0\, (-f(x) + g(x) + f(x)\, g(x)) \\ & + a_1 \left(-\left(\frac{\partial}{\partial x} f(x)\right) - \frac{1}{3} g(x) + \left(\frac{\partial}{\partial x} f(x)\right) g(x) - \frac{1}{3} f(x)\, g(x) \right) \\ & + a_2 \left(\frac{1}{5} f(x) + \frac{1}{9} g(x) - \frac{4}{45} f(x)\, g(x) - \frac{2}{3} \left(\frac{\partial}{\partial x} f(x)\right) g(x) \right) \\ & + a_3 \left(\frac{1}{5} \left(\frac{\partial}{\partial x} f(x)\right) - \frac{1}{27} g(x) + \frac{2}{15} \left(\frac{\partial}{\partial x} f(x)\right) g(x) + \frac{22}{135} f(x)\, g(x) \right) \\ & + a_4 \left(-\frac{1}{25} f(x) + \frac{1}{81} g(x) - \frac{164}{2025} f(x)\, g(x) + \frac{16}{135} \left(\frac{\partial}{\partial x} f(x)\right) g(x) \right) \\ & + a_5 \left(-\frac{1}{25} \left(\frac{\partial}{\partial x} f(x)\right) - \frac{1}{243} g(x) - \frac{244}{2025} \left(\frac{\partial}{\partial x} f(x)\right) g(x) + \frac{4}{1215} f(x)\, g(x) \right) \end{aligned}$$

```
>
```

The maple command collect is for polynomials, not modules, so it doesn't like the form of our basis B:

```
> collect(dependency,B_module);
Error, (in collect) cannot collect, f(x)*g(x)
```

We could replace everything in B by symbols f(x)->b1, diff(f(x),x)->b2, etc., resulting in linear polynomials in the b's, which are the same as a module spanned by the b's. But instead, since the basis is generated by polynomials in f(x), diff(f(x),x), g(x), we do:

```
> B_2 := [f(x),diff(f(x),x),g(x)];
  collect(dependency,B_2,distributed);
```

$$B\_2 := \left[f(x), \frac{\partial}{\partial x} f(x), g(x) \right]$$

$$\left(a_0 - \frac{4}{45} a_2 + \frac{22}{135} a_3 - \frac{1}{3} a_1 - \frac{164}{2025} a_4 + \frac{4}{1215} a_5\right) f(x)\, g(x) + \left(-a_0 - \frac{1}{25} a_4 + \frac{1}{5} a_2\right) f(x) + \left(a_1 + \frac{2}{15} a_3 + \frac{16}{135} a_4 - \frac{2}{3} a_2 - \frac{244}{2025} a_5\right)\left(\frac{\partial}{\partial x} f(x)\right) g(x) + \left(-a_1 - \frac{1}{25} a_5 + \frac{1}{5} a_3\right)\left(\frac{\partial}{\partial x} f(x)\right) + \left(a_0 - \frac{1}{3} a_1 + \frac{1}{9} a_2 - \frac{1}{27} a_3 + \frac{1}{81} a_4 - \frac{1}{243} a_5\right) g(x)$$

```
> B_cofs := coeffs(",B_2);
```

$$B\_cofs := a_0 - \frac{4}{45} a_2 + \frac{22}{135} a_3 - \frac{1}{3} a_1 - \frac{164}{2025} a_4 + \frac{4}{1215} a_5, a_1 + \frac{2}{15} a_3 + \frac{16}{135} a_4 - \frac{2}{3} a_2 - \frac{244}{2025} a_5, -a_0 - \frac{1}{25} a_4 + \frac{1}{5} a_2, -a_1 - \frac{1}{25} a_5 + \frac{1}{5} a_3, a_0 - \frac{1}{3} a_1 + \frac{1}{9} a_2 - \frac{1}{27} a_3 + \frac{1}{81} a_4 - \frac{1}{243} a_5$$

```
> B_sol := solve({B_cofs},{'a[k]'$k=0..5});
```

$$B\_sol := \{ a_3 = \frac{55}{8} a_1, a_2 = \frac{205}{72} a_1, a_4 = \frac{75}{8} a_1, a_5 = \frac{75}{8} a_1, a_0 = \frac{7}{36} a_1, a_1 = a_1 \}$$

So the final differential equation is

```
> hDE := subs(B_sol,dependency_symbollic):
  hDE = 0;
```

$$\frac{7}{36} a_1\, h(x) + a_1 \left(\frac{\partial}{\partial x} h(x)\right) + \frac{205}{72} a_1 \left(\frac{\partial^2}{\partial x^2} h(x)\right) + \frac{55}{8} a_1 \left(\frac{\partial^3}{\partial x^3} h(x)\right) + \frac{75}{8} a_1 \left(\frac{\partial^4}{\partial x^4} h(x)\right) + \frac{75}{8} a_1 \left(\frac{\partial^5}{\partial x^5} h(x)\right) = 0$$

where a[1] is anything at all. Set it to 72 to clear denominators:

```
> hDE := subs(a[1]=72,hDE):
  hDE = 0;
```

$$14\ \mathrm{h}(x)+72\left(\frac{\partial}{\partial x}\ \mathrm{h}(x)\right)+205\left(\frac{\partial^2}{\partial x^2}\ \mathrm{h}(x)\right)+495\left(\frac{\partial^3}{\partial x^3}\ \mathrm{h}(x)\right)+675\left(\frac{\partial^4}{\partial x^4}\ \mathrm{h}(x)\right)$$
$$+675\left(\frac{\partial^5}{\partial x^5}\ \mathrm{h}(x)\right)=0$$

Check it against the explicit solution found earlier:

```
> eval(subs(‘ h‘(x)=h_sol,Diff=diff,hDE));
```

$$-450\left(\frac{1}{25}\_C1\ \%2+\frac{1}{25}\_C2\ \%1\right)\mathbf{e}^{(-1/3\,x)}\ \_C3$$
$$+42\left(-\frac{1}{5}\_C1\ \%1\sqrt{5}+\frac{1}{5}\_C2\ \%2\sqrt{5}\right)\mathbf{e}^{(-1/3\,x)}\ \_C3$$
$$+345\left(\frac{1}{25}\_C1\ \%1\sqrt{5}-\frac{1}{25}\_C2\ \%2\sqrt{5}\right)\mathbf{e}^{(-1/3\,x)}\ \_C3$$
$$+675\left(-\frac{1}{125}\_C1\ \%1\sqrt{5}+\frac{1}{125}\_C2\ \%2\sqrt{5}\right)\mathbf{e}^{(-1/3\,x)}\ \_C3$$
$$-90\left(-\frac{1}{5}\_C1\ \%2-\frac{1}{5}\_C2\ \%1\right)\mathbf{e}^{(-1/3\,x)}\ \_C3$$
$$\%1:=\sin\left(\frac{1}{5}\sqrt{5}\ x\right)$$
$$\%2:=\cos\left(\frac{1}{5}\sqrt{5}\ x\right)$$

```
> simplify(");
```

$$0$$

Chyzak’s routine to compute the same end result as the above, but using Grobner bases.:
First reset all variable names.

```
> f := ’f’; g := ’g’; h := ’h’;
```

$$f := f$$
$$g := g$$
$$h := h$$

```
> with(Groebner); with(Ore_algebra); with(Holonomy);
  with(Mgfun):
```

[*fglm_algo*, *gbasis*, *gsolve*, *hilbertdim*, *hilbertpoly*, *hilbertseries*, *inter_reduce*, *is_finite*, *is_solvable*, *leadcoeff*, *leadmon*, *leadterm*, *normalf*, *pretend_gbasis*, *reduce*, *spoly*, *termorder*, *testorder*, *univpoly*]

[*Ore_to_DESol*, *Ore_to_RESol*, *Ore_to_diff*, *Ore_to_shift*, *annihilators*, *applyopr*, *diff_algebra*, *poly_algebra*, *qshift_algebra*, *rand_skew_poly*, *shift_algebra*, *skew_algebra*, *skew_elim*, *skew_gcdex*, *skew_pdiv*, *skew_power*, *skew_prem*, *skew_product*]

[*algeq_to_dfinite*, *dfinite_add*, *dfinite_mul*, *holon_closure*, *holon_defint*, *holon_defqsum*, *holon_defsum*, *holon_diagonal*, *hypergeom_to_dfinite*, *takayama_algo*]

```
> pol_to_sys(h(x)=-f(x)+g(x)+f(x)*g(x),
        {[f(x), {5*diff(f(x),x$2) + f(x)}],
         [g(x), {3*diff(g(x),x) + g(x)}]});
```

$$\{14\,\mathrm{h}(x) + 72\left(\frac{\partial}{\partial x}\mathrm{h}(x)\right) + 205\,\%1 + 495\left(\frac{\partial^3}{\partial x^3}\mathrm{h}(x)\right) + 675\left(\frac{\partial^4}{\partial x^4}\mathrm{h}(x)\right) + 675\left(\frac{\partial^5}{\partial x^5}\mathrm{h}(x)\right)\}$$

$$\%1 := \frac{\partial^2}{\partial x^2}\mathrm{h}(x)$$

```
>
```